

AutoDoc

Generate documentation from GAP source code

2025.10.16

16 October 2025

Sebastian Gutsche

Max Horn

Sebastian Gutsche

Email: gutsche@mathematik.uni-siegen.de

Homepage: <https://algebra.mathematik.uni-siegen.de/gutsche/>

Address: Department Mathematik

Universität Siegen

Walter-Flex-Straße 3

57072 Siegen

Germany

Max Horn

Email: mhorn@rptu.de

Homepage: <https://www.quendi.de/math>

Address: Fachbereich Mathematik

RPTU Kaiserslautern-Landau

Gottlieb-Daimler-Straße 48

67663 Kaiserslautern

Germany

Abstract

AutoDoc is a GAP package whose purpose is to aid GAP package authors in creating and maintaining the documentation of their packages.

Copyright

© 2012–2022 by Sebastian Gutsche and Max Horn

This package may be distributed under the terms and conditions of the GNU Public License Version 2 or (at your option) any later version.

Acknowledgements

This documentation was prepared using the GAPDoc package [\[LN12\]](#).

Contents

1	Getting started using AutoDoc	4
1.1	Creating a package manual from scratch	4
1.2	Documenting code with AutoDoc	5
1.3	Using AutoDoc in an existing GAPDoc manual	6
1.4	Scaffolds	9
1.5	AutoDoc worksheets	11
2	AutoDoc documentation comments	13
2.1	Documenting declarations	13
2.2	Other documentation comments	15
2.3	Title page commands	19
2.4	Plain text files	20
2.5	Grouping	20
2.6	Level	21
2.7	Markdown-like formatting of text in AutoDoc	22
2.8	Deprecated commands	23
3	AutoDoc worksheets	24
3.1	Worksheets	24
4	AutoDoc	25
4.1	The AutoDoc() function	25
4.2	Examples	29
	References	30
	Index	31

Chapter 1

Getting started using AutoDoc

AutoDoc is a GAP package which is meant to aid GAP package authors in creating and maintaining the documentation of their packages. In this capacity it builds upon the GAPDoc package (see <https://www.gap-system.org/Packages/gapdoc.html>). As such, it is not a replacement for GAPDoc, but rather complements it.

In this chapter we describe how to get started using AutoDoc for your package. First, we explain in Section 1.1 how to write a new package manual from scratch.

Then we show in Section 1.3 how you might benefit from AutoDoc even if you already have a complete manual written using GAPDoc.

In Section 1.4, we explain how you may use AutoDoc to generate a title page and the main XML file for your manual.

Finally, Section 1.5, explains what AutoDoc worksheets are and how to use them.

1.1 Creating a package manual from scratch

Suppose your package is already up and running, but so far has no manual. Then you can rapidly generate a “scaffold” for a package manual using the AutoDoc (4.1.1) command like this, while running GAP from within your package’s directory (the one containing the PackageInfo.g file):

```
LoadPackage( "AutoDoc" );
AutoDoc( rec( scaffold := true ) );
```

This first reads the PackageInfo.g file from the current directory. It extracts information about the package from it (such as its name and version, see Section 1.4.1). It then creates two XML files doc/NAME_OF_YOUR_PACKAGE.xml and doc/title.xml inside the package directory. Finally, it runs GAPDoc on them to produce a nice initial PDF and HTML version of your fresh manual.

To ensure that the GAP help system picks up your package manual, you should also add something like the following to your PackageInfo.g:

```
PackageDoc := rec(
  BookName    := ~.PackageName,
  ArchiveURLSubset := ["doc"],
  HTMLStart   := "doc/chap0.html",
  PDFFile     := "doc/manual.pdf",
  SixFile     := "doc/manual.six",
```

```
LongTitle := ~.Subtitle,
),
```

Congratulations, your package now has a minimal working manual. Of course it will be mostly empty for now, but it already should contain some useful information, based on the data in your `PackageInfo.g`. This includes your package's name, version and description as well as information about its authors. And if you ever change the package data, (e.g. because your email address changed), just re-run the above command to regenerate the two main XML files with the latest information.

Next of course you need to provide actual content (unfortunately, we were not yet able to automate *that* for you, more research on artificial intelligence is required). To add more content, you have several options: You could add further **GAPDoc** XML files containing extra chapters, sections and so on. Or you could use classic **GAPDoc** source comments. For details on either, please refer to (**GAPDoc: Distributing a Document into Several Files**), as well as Section 1.3 of this manual on how to teach the AutoDoc (4.1.1) command to include this extra documentation. Or you could use the special documentation facilities AutoDoc provides (see Section 1.2).

You will probably want to re-run the AutoDoc (4.1.1) command frequently, e.g. whenever you modified your documentation or your `PackageInfo.g`. To make this more convenient and reproducible, we recommend putting its invocation into a file `makedoc.g` in your package directory, with content based on the following example:

```
LoadPackage( "AutoDoc" );
AutoDoc( rec( autodoc := true ) );
QUIT;
```

Then you can regenerate the package manual from the command line with the following command, executed from within in the package directory:

```
gap makedoc.g
```

1.2 Documenting code with AutoDoc

To get one of your global functions, operations, attributes etc. to appear in the package manual, simply insert an AutoDoc comment of the form `#!` directly in front of it. For example:

```
#!
DeclareOperation( "ToricVariety", [ IsConvexObject ] );
```

This tiny change is already sufficient to ensure that the operation appears in the manual. In general, you will want to add further information about the operation, such as in the following example:

```
#! @Arguments conv
#! @Returns a toric variety
#! @Description
#! Creates a toric variety out
#! of the convex object <A>conv</A>.
DeclareOperation( "ToricVariety", [ IsConvexObject ] );
```

For a thorough description of what you can do with **AutoDoc** documentation comments, please refer to chapter 2.

Suppose you have not been using **GAPDoc** before but instead used the process described in section 1.1 to create your manual. Then the following **GAP** command will regenerate the manual and automatically include all newly documented functions, operations etc.:

```
LoadPackage( "AutoDoc" );
AutoDoc( rec( scaffold := true,
              autodoc := true ) );
```

If you are not using the scaffolding feature, e.g. because you already have an existing **GAPDoc** based manual, then you can still use **AutoDoc** documentation comments. Just make sure to first edit the main XML file of your documentation, and insert the line

```
<#Include SYSTEM "_AutoDocMainFile.xml">
```

in a suitable place. This means that you can mix **AutoDoc** documentation comment freely with your existing documentation; you can even still make use of any existing **GAPDoc** documentation comments in your code. The following command should be useful for you in this case; it still scans the package code for **AutoDoc** documentation comments and the runs **GAPDoc** to produce HTML and PDF output, but does not touch your documentation XML files otherwise.

```
LoadPackage( "AutoDoc" );
AutoDoc( rec( autodoc := true ) );
```

1.3 Using AutoDoc in an existing GAPDoc manual

Even if you already have an existing **GAPDoc** manual, it might be interesting for you to use **AutoDoc** for two purposes:

First off, with **AutoDoc** it is very convenient to regenerate your documentation.

Secondly, the scaffolding feature which generates a title page with all the metadata of your package in a uniform way is very handy. The somewhat tedious process of keeping your title page in sync with your `PackageInfo.g` is fully automated this way (including the correct version, release data, author information and so on).

There are various examples of packages using **AutoDoc** for this purpose only, e.g. `io` and `orb`.

1.3.1 Using AutoDoc on a complete GAPDoc manual

Suppose you already have a complete XML manual, with some main and title XML files and some documentation for operations distributed over all your `.g`, `.gd`, and `.gi` files. Suppose the main XML file is named `PACKAGENAME.xml` and is in the `/doc` subdirectory of your package. Then you can rebuild your manual by executing the following two **GAP** commands from a **GAP** session started in the root directory of your package:

```
LoadPackage( "AutoDoc" );
AutoDoc( );
```

In contrast, the `RingsForHomalg` package currently uses essentially the following code in its `makedoc.g` file to achieve the same result:

```
LoadPackage( "GAPDoc" );
SetGapDocLaTeXOptions( "utf8" );
bib := ParseBibFiles( "doc/RingsForHomalg.bib" );
WriteBibXMLExtFile( "doc/RingsForHomalgBib.xml", bib );
list := [
    "../gap/RingsForHomalg.gd",
    "../gap/RingsForHomalg.gi",
    "../gap/Singular.gi",
    "../gap/SingularBasic.gi",
    "../examples/RingConstructionsExternalGAP.g",
    "../examples/RingConstructionsSingular.g",
    "../examples/RingConstructionsMAGMA.g",
    "../examples/RingConstructionsMacaulay2.g",
    "../examples/RingConstructionsSage.g",
    "../examples/RingConstructionsMaple.g",
];
MakeGAPDocDoc( "doc", "RingsForHomalg", list, "RingsForHomalg" );
GAPDocManualLab( "RingsForHomalg" );
```

Note that in particular, you do not have to worry about keeping a list of your implementation files up-to-date.

But there is more. `AutoDoc` can create `.tst` files from the examples in your manual to test your package. This can be achieved via

```
LoadPackage( "AutoDoc" );
AutoDoc( rec( extract_examples := true ) );
```

Now files `PACKAGENAME01.tst`, `PACKAGENAME02.tst` and so appear in the `tst/` subdirectory of your package, and can be tested as usual using `Test` (**Reference: Test**) respectively `TestDirectory` (**Reference: TestDirectory**).

1.3.2 Setting different `GAPDoc` options

Sometimes, the default values for the `GAPDoc` command used by `AutoDoc` may not be suitable for your manual.

Suppose your main XML file is *not* named `PACKAGENAME.xml`, but rather something else, e.g. `main.xml`. Then you can tell `AutoDoc` to use this file as the main XML file via

```
LoadPackage( "AutoDoc" );
AutoDoc( rec( gapdoc := rec( main := "main" ) ) );
```

As explained above, by default `AutoDoc` scans all `.g`, `.gd` and `.gi` files it can find inside of your package root directory, and in the subdirectories `gap`, `lib`, `examples` and `examples/doc` as well. If you keep source files with documentation in other directories, you can adjust the list of directories `AutoDoc` scans via the `scan_dirs` option. The following example illustrates this by instructing `AutoDoc` to only search in the subdirectory `package_sources` of the packages root directory.

```
LoadPackage( "AutoDoc" );
AutoDoc( rec( gapdoc := rec( scan_dirs := [ "package_sources" ] ) ) );
```

You can also specify an explicit list of files containing documentation, which will be searched in addition to any files located within the scan directories:

```
LoadPackage( "AutoDoc" );
AutoDoc( rec( gapdoc := rec( files := [ "path/to/some/hidden/file.gds" ] ) ) );
```

Giving such a file does not prevent the standard `scan_dirs` from being scanned for other files.

Next, **GAPDoc** supports the documentation to be built with relative paths. This means, links to manuals of other packages or the **GAP** library will not be absolute, but relative from your documentation. This can be particularly useful if you want to build a release tarball or move your **GAP** installation around later. Suppose you are starting **GAP** in the root path of your package as always, and the standard call of `AutoDoc` (4.1.1) will then build the documentation in the `doc` subdirectory of your package. From this directory, the gap root directory has the relative path `../../..`. Then you can enable the relative paths by

```
LoadPackage( "AutoDoc" );
AutoDoc( rec( gapdoc := rec( gap_root_relative_path := "../../.." ) ) );
```

or, since `../../..` is the standard option for `gap_root_relative_path`, by

```
LoadPackage( "AutoDoc" );
AutoDoc( rec( gapdoc := rec( gap_root_relative_path := true ) ) );
```

1.3.3 Checklist for converting an existing **GAPDoc** manual to use **AutoDoc**

Here is a checklist for authors of a package `PackageName`, which already has a *GAPDoc* based manual, who wish to use **AutoDoc** to build the manual from now on. We assume that the manual is currently built by reading a file `makedoc.g` and that the main `.xml` file is named `manual.xml`.

The files `PackageInfo.g`, `makedoc.g`, `doc/title.xml` and `doc/PackageName.xml` (if it exists) will be altered by this procedure, so it may be wise to keep backup copies.

You should have copies of the **AutoDoc** files `PackageInfo.g` and `makedoc.g` to hand when reading these instructions.

- Copy `AutoDoc/makedoc.g` to `PackageName/makedoc.g`.
- Edit `PackageName/makedoc.g` as follows.
 - Change the header comment to match other files in your package.
 - After `LoadPackage("AutoDoc");` add `LoadPackage("PackageName");`.
 - In the `AutoDoc` record delete `autodoc := true;`.
 - In the scaffold record change the `includes` list to be the list of your `.xml` files that are contained in `manual.xml`.

- If you do not have a bibliography you may delete the `bib := "bib.xml"`, field in the scaffold. Otherwise, edit the file name if you have a different file. If you only have a `.bib` file (`manual.bib` or `bib.xml.bib` say) you should rename this file `PackageName.bib`.
- In the `LaTeXOptions` record, which is in the `gapdoc` record, enter any `LATEX` newcommands previously in `manual.xml`. (If there are none you may safely delete this record.) To illustrate this option, the `AutoDoc` file `makedoc.g` defines the command `\bbZ` by `\newcommand{\bbZ}{\mathbb{Z}}`, which may be used to produce the `LATEX` formula $f : \mathbb{Z}^2 \rightarrow \mathbb{Z}$. However, note that this only works in the PDF version of the file.
- Now edit `PackageName/PackageInfo.g` as follows.
 - Delete any `PKGVERSIONDATA` chunk that may be there. One reason for converting your manual to use `AutoDoc` is to stop using entities such as `PACKAGENAMEVERSION`.
 - Copy the `AutoDoc` record from `AutoDoc/PackageInfo.g` to the end of your file (just before the final `")) ; "`.
 - Replace the `Copyright`, `Abstract` and `Acknowledgements` fields of the `TitlePage` record with the corresponding material from your `manual.xml`. (If you do not have an abstract, then delete the `Abstract` field, etc.)
 - In the `entities` record enter any entities previously stored in your `manual.xml`. (Again, if you have none, you may safely delete this record.) To illustrate this option the `AutoDoc` file `PackageInfo.g` defines entities for the names of packages `io` and `PackageName`.
- If you are using a GitHub repository, as well as running `"git add"` on files `makedoc.g`, `PackageInfo.g` and `doc/PackageName.bib`, you should run `"git rm -f"` on files `doc/manual.xml`, and `doc/title.xml`.

You should now be ready to run `GAP` and `Read("makedoc.g")` ; in your package root directory.

1.4 Scaffolds

1.4.1 Generating a title page

For most (if not all) `GAP` packages, the title page of the package manual lists information such as the release date, version, names and contact details of the authors, and so on. All this data is also contained in your `PackageInfo.g`, and whenever you make a change to that file, there is a risk that you forget to update your manual to match. And even if you don't forget it, you of course have to spend some time to adjust the manual. `GAPDoc` can help to a degree with this via entities. Thus, you will sometimes see code like this in `PackageInfo.g` files:

```
Version      := "1.2.3",
Date        := "20/01/2015",
## <#GAPDoc Label="PKGVERSIONDATA">
## <!ENTITY VERSION "1.2.3">
## <!ENTITY RELEASEDATE "20 January 2015">
## <!ENTITY RELEASEYEAR "2015">
## <#/GAPDoc>
```

However, it is still easy to forget both of these versions. And it doesn't solve the problem of updating package authors addresses. Neither of these is a big issue, of course, but there have been plenty examples in the past where people forget either of these two things, and it can be slightly embarrassing. It may even require you to make a new release just to fix the issue, which in our opinion is a sad waste of your valuable time.

So instead of worrying about manually synchronising these things, you can instruct **AutoDoc** to generate a title page for your manual based on the information in your `PackageInfo.g`. The following commands do just that (in addition to building your manual), by generating a file called `doc/title.xml`.

```
LoadPackage( "AutoDoc" );
AutoDoc( rec( scaffold := rec( MainPage := false ) ) );
```

Note that this only outputs `doc/title.xml` but does not touch any other files of your documentation. In particular, you need to explicitly include `doc/title.xml` from your main XML file.

However, you can also tell **AutoDoc** to maintain the main XML file for you, in which case this is automatic. In fact, this is the default if you enable scaffolding; the above example command explicitly told **AutoDoc** not to generate a main page.

1.4.2 Generating the main XML file

The following generates a main XML file for your documentation in addition to the title page. The main XML file includes the title page by default, as well as any documentation generated from **AutoDoc** documentation comments.

```
LoadPackage( "AutoDoc" );
AutoDoc( rec( scaffold := true ) );
```

You can instruct **AutoDoc** to include additional XML files by giving it a list of filenames, as in the following example:

```
LoadPackage( "AutoDoc" );
AutoDoc(rec(
  scaffold := rec(
    includes := [ "somefile.xml", "anotherfile.xml" ]
  )
));
```

For more information, please consult the documentation of the **AutoDoc** (4.1.1) function.

1.4.3 What data is used from `PackageInfo.g`?

AutoDoc can use data from `PackageInfo.g` in order to generate a title page. Specifically, the following components of the package info record are taken into account:

PackageName

This is used to set the `<Title>` element of the title page.

Subtitle

This is used to set the `<Subtitle>` element of the title page.

Version

This is used to set the <Version> element of the title page, with the string “Version ” prepended.

Date This is used to set the <Date> element of the title page.

Persons

This is used to generate <Author> elements in the generated title page.

PackageDoc

This is a record (or a list of records) which is used to tell the GAP help system about the package manual. Currently AutoDoc extracts the value of the PackageDoc.BookName component and then passes that on to GAPDoc when creating the HTML, PDF and text versions of the manual.

AutoDoc

This is a record which can be used to control the scaffolding performed by AutoDoc, specifically to provide extra information for the title page. For example, you can set AutoDoc.TitlePage.Copyright to a string which will then be inserted on the generated title page. Using this method you can customize the following title page elements: TitleComment, Abstract, Copyright, Acknowledgements and Colophon.

Note that AutoDoc.TitlePage behaves exactly the same as the scaffold.TitlePage parameter of the AutoDoc (4.1.1) function.

1.5 AutoDoc worksheets

AutoDoc worksheets can be used to create HTML and PDF documents using AutoDoc syntax and possibly including GAP examples and implementations without having them associated to a package. A file for a worksheet could look like this:

```
#! @Title My first worksheet
#! @Author Charlie Brown

#! @Chapter Some groups

#! @BeginExample
S3 := SymmetricGroup( 3 );
S4 := SymmetricGroup( 4 );
#! @EndExample
```

Now, one can create a PDF and HTML document, like a package documentation out of it. Suppose the document above is saved as worksheet.g. Then, when GAP is started in the directory of this file, the command

```
AutoDocWorksheet( "worksheet.g" );
```

will create a subdirectory called doc of the current directory in which it will create the documentation. There are several options to configure the output of the worksheet command, which are identical to the options of the AutoDoc (4.1.1) command. It is even possible to test the examples in the worksheet using the extract_examples option of the AutoDoc (4.1.1) command.

Since the worksheets do not have a `PackageInfo.g` to extract information, all possible tags that GAPDoc supports for the title page can be set into the document. A fully typed title page can look like this:

```
#! @Title My first worksheet
#! @Subtitle Some small examples
#! @Author Charlie Brown

#! @Version 0.1
#! @TitleComment Some worksheet
#! @Date 01/01/2016
#! @Address TU Kaiserslautern
#! @Abstract
#! A worksheet showing some small examples about groups.
#! @Copyright 2016 Charlie Brown
#! @Acknowledgements Woodstock
#! @Colophon Some colophon

#! @Chapter Some groups

#! @BeginExample
S3 := SymmetricGroup( 3 );;
S4 := SymmetricGroup( 4 );;
#! @EndExample
```

Chapter 2

AutoDoc documentation comments

You can document declarations of global functions and variables, operations, attributes etc. by inserting *AutoDoc* comments into your sources before these declaration. An *AutoDoc* comment always starts with `#!`. This is also the smallest possible *AutoDoc* command. If you want your declaration documented, just write `#!` at the line before the documentation. For example:

```
#!  
DeclareOperation( "AnOperation",  
                 [ IsList ] );
```

This will produce a manual entry for the operation `AnOperation`.

Inside of *AutoDoc* comments, *AutoDoc commands* starting with `@` can be used to control the output *AutoDoc* produces.

2.1 Documenting declarations

In the bare form above, the manual entry for `AnOperation` will not contain much more than the name of the operation. In order to change this, there are several commands you can put into the *AutoDoc* comment before the declaration. Currently, the following commands are provided:

2.1.1 `@Description descr`

Adds the text in the following lines of the *AutoDoc* to the description of the declaration in the manual. Lines are until the next *AutoDoc* command or until the declaration is reached.

2.1.2 `@Returns ret_val`

The string `ret_val` is added to the documentation, with the text “Returns: ” put in front of it. This should usually give a brief hint about the type or meaning of the value returned by the documented function.

2.1.3 `@Arguments args`

The string `args` contains a description of the arguments the function expects, including optional parts, which are denoted by square brackets. The argument names can be separated by whitespace, commas

or square brackets for the optional arguments, like “grp[, elm]” or “xx[y[z]]”. If GAP options are used, this can be followed by a colon : and one or more assignments, like “n[, r]: tries := 100”.

2.1.4 @Group *grpname*

Adds the following method to a group with the given name. See section 2.5 for more information about groups.

2.1.5 @Label *label*

Adds label to the function as label. If this is not specified, then for declarations that involve a list of input filters (as is the case for DeclareOperation, DeclareAttribute, etc.), a default label is generated from this filter list.

```
#! @Label testlabel
DeclareProperty( "AProperty",
                IsObject );
```

leads to this:

2.1.6 AProperty (testlabel)

▷ AProperty(*arg*) (property)
Returns: true or false
while

```
#!
DeclareProperty( "AProperty",
                IsObject );
```

leads to this:

2.1.7 AProperty (for IsObject)

▷ AProperty(*arg*) (property)
Returns: true or false

2.1.8 @ChapterInfo *chapter, section*

Adds the entry to the given chapter and section. Here, *chapter* and *section* are the respective titles. As an example, a full AutoDoc comment with all options could look like this:

```
#! @Description
#! Computes the list of lists of degrees of ordinary characters
#! associated to the $p$-blocks of the group $G$
#! with $p$-modular character table <A>modtbl</A>
#! and underlying ordinary character table ‘ordtbl’.
#! @Returns a list
#! @Arguments modtbl
#! @Group CharacterDegreesOfBlocks
```

```

#! @Label chardegblocks
#! @ChapterInfo Blocks, Attributes
DeclareAttribute( "CharacterDegreesOfBlocks",
                 IsBrauerTable );

```

2.2 Other documentation comments

There are also some commands which can be used in AutoDoc comments that are not associated to any declaration. This is useful for additional text in your documentation, examples, mathematical chapters, etc..

2.2.1 @Chapter *name*

Sets the active chapter, all subsequent functions which do not have an explicit chapter declared in their AutoDoc comment via @ChapterInfo will be added to this chapter. Also all text comments, i.e. lines that begin with #! without a command, and which do not follow after @Description, will be added to the chapter as regular text. Additionally, the chapters label will be set to *Chapter_name*. Example:

```

#! @Chapter My chapter
#! This is my chapter.
#! I document my stuff in it.

```

The @ChapterLabel *label* command can be used to set the label of the chapter to *Chapter_label* instead of *Chapter_name*. Additionally, the chapter will be stored as *_Chapter_label.xml*. The @ChapterTitle *title* command can be used to set a heading for the chapter that is different from *name*. Note that the title does not affect the label. If you use all three commands, i.e.,

```

#! @Chapter name
#! @ChapterLabel label
#! @ChapterTitle title

```

title is used for the headline, *label* for cross-referencing, and *name* for setting the same chapter as active chapter again.

2.2.2 @Section *name*

Sets an active section like @Chapter sets an active chapter. The section automatically ends with the next @Section or @Chapter command.

```

#! @Section My first manual section
#! In this section I am going to document my first method.

```

The @SectionLabel *label* command can be used to set the label of the section to *Section_label* instead of *Chapter_chaptername_Section_name*. The @SectionTitle *title* command can be used to set a heading for the section that is different from *name*.

2.2.3 @Subsection *name*

Sets an active subsection like @Section sets an active section. The subsection automatically ends with the next @Subsection, @Section or @Chapter command. It also ends with the next documented function. Indeed, internally each function “manpage” is treated like a subsection.

```
#! @Subsection My first manual subsection
#! In this subsection I am going to document my first example.
```

The @SubsectionLabel *label* command can be used to set the label of the subsection to *Subsection_label* instead of *Chapter_chaptername_Section_sectionname_Subsection_name*. The @SubsectionTitle *title* command can be used to set a heading for the subsection that is different from *name*.

2.2.4 @BeginGroup [*grpname*]

Starts a group. All following documented declarations without an explicit @Group command are grouped together in the same group with the given name. If no name is given, then a new nameless group is generated. The effect of this command is ended when an @EndGroup command is reached.

See section 2.5 for more information about groups.

2.2.5 @EndGroup

Ends the current group.

```
#! @BeginGroup MyGroup
#!
DeclareAttribute( "GroupedAttribute",
                  IsList );

DeclareOperation( "NonGroupedOperation",
                  [ IsObject ] );

#!
DeclareOperation( "GroupedOperation",
                  [ IsList, IsRubbish ] );
#! @EndGroup
```

2.2.6 @GroupTitle *title*

Sets the subsection heading for the current group to *title*. In the absence of any @GroupTitle command, the heading will be the name of the first entry in the group. See 2.5 for more information.

2.2.7 @Level *lvl*

Sets the current level of the documentation. All items created after this, chapters, sections, and items, are given the level *lvl*, until the @ResetLevel command resets the level to 0 or another level is set.

See section 2.6 for more information about levels.

2.2.8 @ResetLevel

Resets the current level to 0.

2.2.9 @BeginExample and @EndExample

@BeginExample marks the start of an example to be put into the manual. It differs from GAPDoc's <Example> (see (**GAPDoc: Log**)), in that it expects actual code (not in a comment) interspersed with comments, to allow for examples files that can be both executed by GAP, and parsed by AutoDoc. To achieve this, GAP commands are not preceded by a comment, while output has to be preceded by an AutoDoc comment. The gap> prompt for the display in the manual is added by AutoDoc. @EndExample ends the example block.

To illustrate this command, consider this input:

```
#! @BeginExample
S5 := SymmetricGroup(5);
#! Sym( [ 1 .. 5 ] )
Order(S5);
#! 120
#! @EndExample
```

This results in the following output:

Example

```
gap> S5 := SymmetricGroup(5);
Sym( [ 1 .. 5 ] )
gap> Order(S5);
120
```

The AutoDoc command @Example is an alias of @BeginExample.

2.2.10 @BeginExampleSession and @EndExampleSession

@BeginExampleSession marks the start of an example to be put into the manual, while @EndExampleSession ends the example block. It is the direct analog of GAPDoc's <Example> (see (**GAPDoc: Log**)).

To illustrate this command, consider this input:

```
#! @BeginExampleSession
#! gap> S5 := SymmetricGroup(5);
#! Sym( [ 1 .. 5 ] )
#! gap> Order(S5);
#! 120
#! @EndExampleSession
```

This results in the following output:

Example

```
gap> S5 := SymmetricGroup(5);
Sym( [ 1 .. 5 ] )
gap> Order(S5);
120
```

It inserts an example into the manual just as `@Example` would do, but all lines are commented and therefore not executed when the file is read. All lines that should be part of the example displayed in the manual have to start with an **AutoDoc** comment (`#!`). The comment will be removed, and, if the following character is a space, this space will also be removed. There is never more than one space removed. To ensure examples are correctly colored in the manual, there should be exactly one space between `#!` and the `gap>` prompt. The **AutoDoc** command `@ExampleSession` is an alias of `@BeginExampleSession`.

2.2.11 `@BeginLog` and `@EndLog`

Works just like the `@BeginExample` command, but the example will not be tested. See the **GAPDoc** manual for more information. The **AutoDoc** command `@Log` is an alias of `@BeginLog`.

2.2.12 `@BeginLogSession` and `@EndLogSession`

Works just like the `@BeginExampleSession` command, but the example will not be tested if manual examples are run. It is the direct analog of **GAPDoc**'s `<Log>` (see (**GAPDoc: Log**)). The **AutoDoc** command `@LogSession` is an alias of `@BeginLogSession`.

2.2.13 `@DoNotReadRestOfFile`

Prevents the rest of the file from being read by the parser. Useful for unfinished or temporary files.

```
#! This will appear in the manual

#! @DoNotReadRestOfFile

#! This will not appear in the manual.
```

2.2.14 `@BeginChunk name`, `@EndChunk`, and `@InsertChunk name`

Text inside a `@BeginChunk` / `@EndChunk` part will not be inserted into the final documentation directly. Instead, the text is stored in an internal buffer. That chunk of text can then later on be inserted in any other place by using the `@InsertChunk name` command. If you do not provide an `@EndChunk`, the chunk ends at the end of the file.

```
#! @BeginChunk MyChunk
#! Hello, world.
#! @EndChunk

#! @InsertChunk MyChunk
## The text "Hello, world." is inserted right before this.
```

You can use this to define an example like this in one file:

```
#! @BeginChunk Example_Symmetric_Group
#! @BeginExample
S5 := SymmetricGroup(5);
#! Sym( [ 1 .. 5 ] )
Order(S5);
```

```

#! 120
#! @EndExample
#! @EndChunk

```

And then later, insert the example in a different file, like this:

```

#! @InsertChunk Example_Symmetric_Group

```

2.2.15 @BeginCode *name*, @EndCode, and @InsertCode *name*

Inserts the text between @BeginCode and @EndCode verbatim at the point where @InsertCode is called. This is useful to insert code excerpts directly into the manual.

```

#! @BeginCode Increment
i := i + 1;
#! @EndCode

#! @InsertCode Increment
## Code is inserted here.

```

2.2.16 @LatexOnly *text*, @BeginLatexOnly, and @EndLatexOnly

Code inserted between @BeginLatexOnly and @EndLatexOnly or after @LatexOnly is only inserted in the PDF version of the manual or worksheet. It can hold arbitrary L^AT_EX-commands.

```

#! @BeginLatexOnly
#! \include{picture.tex}
#! @EndLatexOnly

#! @LatexOnly \include{picture.tex}

```

2.2.17 @NotLatex *text*, @BeginNotLatex, and @EndNotLatex

Code inserted between @BeginNotLatex and @EndNotLatex or after @NotLatex is inserted in the HTML and text versions of the manual or worksheet, but not in the PDF version.

```

#! @BeginNotLatex
#! For further information see the PDF version of this manual.
#! @EndNotLatex

#! @NotLatex For further information see the PDF version of this manual.

```

2.3 Title page commands

The following commands can be used to add the corresponding parts to the title page of the document which generated by AutoDoc if scaffolding is enabled.

- @Title

- `@Subtitle`
- `@Version`
- `@TitleComment`
- `@Author`
- `@Date`
- `@Address`
- `@Abstract`
- `@Copyright`
- `@Acknowledgements`
- `@Colophon`

Those add the following lines at the corresponding point of the title page. Please note that many of those things can be (better) extracted from the `PackageInfo.g`. In case you set some of those, the extracted or in scaffold defined items will be overwritten. While this is not very useful for documenting packages, they are necessary for worksheets created with `AutoDocWorksheet` (3.1.1), since worksheets do not have a `PackageInfo.g` file from which this information could be extracted.

2.4 Plain text files

Files that have the suffix `.autodoc` and are listed in the `autodoc.files` option of `AutoDoc` (4.1.1), resp. are contained in one of the directories listed in `autodoc.scan_dirs`, are treated as `AutoDoc` plain text files. These work exactly like `AutoDoc` comments, except that lines do not need to (and in fact, should not) start with `#!`.

2.5 Grouping

In `GAPDoc`, it is possible to make groups of manual items, i.e., when documenting a function, operation, etc., it is possible to group them into suitable chunks. This can be particularly useful if there are several definitions of an operation with several different argument types, all doing more or less the same to the arguments. Then their manual items can be grouped, sharing the same description and return type information. You can give a heading to the group in the manual with the `@GroupTitle` command; if that is not supplied, then the heading of the first manual item in the group will be used as the heading.

Note that group names are globally unique throughout the whole manual. That is, groups with the same name are in fact merged into a single group, even if they were declared in different source files. Thus you can have multiple `@BeginGroup` / `@EndGroup` pairs using the same group name, in different places, and these all will refer to the same group.

Moreover, this means that you can add items to a group via the `@Group` command in the `AutoDoc` comment of an arbitrary declaration, at any time.

The following code

```

#! @BeginGroup Group1
#! @GroupTitle A family of operations

#! @Description
#! First sentence.
DeclareOperation( "FirstOperation", [ IsInt ] );

#! @Description
#! Second sentence.
DeclareOperation( "SecondOperation", [ IsInt, IsGroup ] );

#! @EndGroup

## .. Stuff ..

#! @Description
#! Third sentence.
#! @Group Group1
KeyDependentOperation( "ThirdOperation", IsGroup, IsInt, "prime );

```

produces the following:

2.5.1 A family of operations

- ▷ FirstOperation(arg) (operation)
- ▷ SecondOperation(arg1, arg2) (operation)
- ▷ ThirdOperation(arg1, arg2) (operation)

First sentence. Second sentence. Third sentence.

2.6 Level

Levels can be set to not write certain parts in the manual by default. Every entry has by default the level 0. The command @Level can be used to set the level of the following part to a higher level, for example 1, and prevent it from being printed to the manual by default. However, if one sets the level to a higher value in the autodoc option of **AutoDoc**, the parts will be included in the manual at the specific place.

```

#! This text will be printed to the manual.
#! @Level 1
#! This text will be printed to the manual if created with level 1 or higher.
#! @Level 2
#! This text will be printed to the manual if created with level 2 or higher.
#! @ResetLevel
#! This text will be printed to the manual.

```

2.7 Markdown-like formatting of text in AutoDoc

AutoDoc has some convenient ways to insert special format into text, like math formulas and lists. The syntax for them are inspired by Markdown and $\text{\LaTeX}$, but do not follow them strictly. Neither are all features of the Markdown language supported. The following subsections describe what is possible.

2.7.1 Lists

One can create lists of items by beginning a new line with `*`, `+`, `-`, followed by one space. The first item starts the list. When items are longer than one line, the following lines have to be indented by at least two spaces. The list ends when a line which does not start a new item is not indented by two spaces. Of course lists can be nested. Here is an example:

```
#! The list starts in the next line
#! * item 1
#! * item 2
#!   which is a bit longer
#!   * and also contains a nested list
#!     * with two items
#! * item 3 of the outer list
#! This does not belong to the list anymore.
```

This is the output:

The list starts in the next line

- item 1
- item 2 which is a bit longer
 - and also contains a nested list
 - with two items
- item 3 of the outer list

This does not belong to the list anymore.

The `*`, `-`, and `+` are fully interchangeable and can even be used mixed, but this is not recommended.

2.7.2 Math modes

One can start an inline formula with a `$`, and also end it with `$`, just like in $\text{\LaTeX}$. This will translate into GAPDocs inline math environment. For display mode one can use `$$`, also like $\text{\LaTeX}$.

```
#! This is an inline formula: $1+1 = 2$.
#! This is a display formula:
#! $$ \sum_{i=1}^n i. $$
```

produces the following output:

This is an inline formula: $1 + 1 = 2$. This is a display formula:

$$\sum_{i=1}^n i.$$

2.7.3 Emphasize

One can emphasize text by using two asterisks (**) or two underscores (__) at the beginning and the end of the text which should be emphasized. Example:

```
#! **This** is very important.
#! This is __also important__.
#! **Naturally, more than one line
#! can be important.**
```

This produces the following output:

This is very important. This is also important. Naturally, more than one line can be important.

2.7.4 Inline code

One can mark inline code snippets by using backticks (`) at the beginning and the end of the text which should be marked as code. Example:

```
#! Call function 'foobar()' at the start.
```

This produces the following output:

Call function `foobar()` at the start.

2.8 Deprecated commands

The following commands used to be supported, but are not anymore.

@EndSection

You can simply remove any use of this, AutoDoc ends sections automatically at the start of any new section or chapter.

@EndSubsection

You can simply remove any use of this, AutoDoc ends subsections automatically at the start of any new subsection, section or chapter.

@BeginAutoDoc **and** @EndAutoDoc

It suffices to prepend each declaration that is meant to be appear in the manual with a minimal AutoDoc comment #!.

@BeginSystem *name*, @EndSystem, **and** @InsertSystem *name*

Please use the chunk commands from subsection [2.2.14](#) instead.

@AutoDocPlainText **and** @EndAutoDocPlainText

Use .autodoc files or AutoDoc comments instead.

Chapter 3

AutoDoc worksheets

3.1 Worksheets

3.1.1 AutoDocWorksheet

▷ `AutoDocWorksheet(list_of_filenames: options)` (function)

The intention of these function is to create stand-alone pdf and html files using AutoDoc without having them associated to a package. It uses the same optional records as the **AutoDoc** command itself, but instead of a package name there should be a filename or a list of filenames containing AutoDoc text from which the documents are created. Please see the **AutoDoc** command for more information about this and have a look at [1.5](#) for a simple worksheet example.

Chapter 4

AutoDoc

4.1 The AutoDoc() function

4.1.1 AutoDoc

▷ `AutoDoc([packageOrDirectory][,] [optrec])` (function)

Returns: nothing

This is the main function of the `AutoDoc` package. It can perform any combination of the following tasks:

1. It can (re)generate a scaffold for your package manual. That is, it can produce two XML files in `GAPDoc` format to be used as part of your manual: First, a file named `doc/PACKAGENAME.xml` (with your package's name substituted) which is used as main XML file for the package manual, i.e. this file sets the XML doctype and defines various XML entities, includes other XML files (both those generated by `AutoDoc` as well as additional files created by other means), tells `GAPDoc` to generate a table of contents and an index, and more. Secondly, it creates a file `doc/title.xml` containing a title page for your documentation, with information about your package (name, description, version), its authors and more, based on the data in your `PackageInfo.g`.
2. It can scan your package for `AutoDoc` based documentation (by using `AutoDoc` tags and the `Autodoc` command. This will produce further XML files to be used as part of the package manual.
3. It can use `GAPDoc` to generate PDF, text and HTML (with MathJaX enabled) documentation from the `GAPDoc` XML files it generated as well as additional such files provided by you. For this, it invokes `MakeGAPDocDoc` (**GAPDoc: MakeGAPDocDoc**) to convert the XML sources, and it also instructs `GAPDoc` to copy supplementary files (such as CSS style files) into your `doc` directory (see `CopyHTMLStyleFiles` (**GAPDoc: CopyHTMLStyleFiles**)).

For more information and some examples, please refer to Chapter 1.

The parameters have the following meanings:

packageOrDirectory

The purpose of this parameter is twofold: to determine the package directory in which `AutoDoc` will operate, and to find the metadata concerning the package being documented. The parameter is either a string, an `IsDirectory` object, or omitted. If it is a string, `AutoDoc` interprets

it as the name of a package, uses the metadata of the first package known to **GAP** with that name, and uses the `InstallationPath` specified in that metadata as the package directory. If `packageOrDirectory` is an `IsDirectory` object, this is used as package directory; if the argument is omitted, then the current directory is used. In the last two cases, the specified directory must contain a `PackageInfo.g` file, and **AutoDoc** extracts all needed metadata from that. The `IsDirectory` form is for example useful if you have multiple versions of the package around and want to make sure the documentation of the correct version is built.

Note that when using `AutoDocWorksheet` (see 3.1), there is no parameter corresponding to `packageOrDirectory` and so the “package directory” is always the current directory, and there is no metadata.

optrec

optrec can be a record with some additional options. The following are currently supported:

dir This should either be a string, in which case it must be a path *relative* to the specified package directory, or a `Directory()` object. (Thus, the only way to designate an absolute directory is with a `Directory()` object.) This option specifies where the package documentation (e.g. the **GAPDoc** XML or the manual PDF, etc.) files are stored and/or will be generated.

Default value: `"doc/"`.

scaffold

This controls whether and how to generate scaffold XML files for the package documentation.

The value should be either `true`, `false` or a record. If it is a record or `true` (the latter is equivalent to specifying an empty record), then this feature is enabled. It is also enabled if `opt.scaffold` is missing but the package’s info record in `PackageInfo.g` has an **AutoDoc** entry. In all other cases (in particular if `opt.scaffold` is `false`), scaffolding is disabled.

If scaffolding is enabled, and `PackageInfo.AutoDoc` exists, then it is assumed to be a record, and its contents are used as default values for the scaffold settings.

If `opt.scaffold` is a record, it may contain the following entries.

includes

A list of XML files to be included in the body of the main XML file. If you specify this list and also are using **AutoDoc** to document your operations with **AutoDoc** comments, you can add `_AutoDocMainFile.xml` to this list to control at which point the documentation produced by **AutoDoc** is inserted. If you do not do this, it will be added after the last of your own XML files.

index

By default, the scaffold creates an index. If you do not want an index, set this to `false`.

appendix

This entry is similar to `opt.scaffold.includes` but is used to specify files to include after the main body of the manual, i.e. typically appendices.

bib

The name of a bibliography file, in BibTeX or XML format. If this key is not set, but

there is a file `doc/PACKAGENAME.bib` then it is assumed that you want to use this as your bibliography.

entities

A record whose keys are taken as entity names, set to the corresponding (string) values. For example, if you pass the record “SomePackage”,

```
rec( SomePackage := "<Package>SomePackage</Package>",
      RELEASEYEAR := "2015" )
```

then the following entity definitions are added to the XML preamble:

```
<!ENTITY SomePackage '<Package>SomePackage</Package>''>
<!ENTITY RELEASEYEAR '2015'>
```

This allows you to write “&SomePackage;” and “&RELEASEYEAR;” in your documentation, which will be replaced by respective values specified in the entities definition.

TitlePage

A record whose entries are used to embellish the generated title page for the package manual with extra information, such as a copyright statement or acknowledgments. To this end, the names of the record components are used as XML element names, and the values of the components are outputted as content of these XML elements. For example, you could pass the following record to set a custom acknowledgements text:

```
rec( Acknowledgements := "Many thanks to ..." )
```

For a list of valid entries in the title page, please refer to the **GAPDoc** manual, specifically section (**GAPDoc: TitlePage**).

MainPage

If scaffolding is enabled, by default a main XML file is generated (this is the file which contains the XML doctype and more). If you do not want this (e.g. because you have a handwritten main XML file), but still want **AutoDoc** to generate a title page for you, you can set this option to `false`

document_class

Sets the document class of the resulting PDF. The value can either be a string which has to be the name of the new document class, a list containing this string, or a list of two strings. Then the first one has to be the document class name, the second one the option string (contained in `[ ]`) in $\text{\LaTeX}$.

latex_header_file

Replaces the standard header from **GAPDoc** completely with the header in this $\text{\LaTeX}$ file. Please be careful here, and look at **GAPDoc**’s `latexheader.tex` file for an example.

autodoc

This controls whether and how to generate additional XML documentation files by scanning for **AutoDoc** documentation comments.

The value should be either `true`, `false` or a record. If it is a record or `true` (the latter is equivalent to specifying an empty record), then this feature is enabled. It is also enabled if `opt.autodoc` is missing but the package depends (directly) on the **AutoDoc** package. In all other cases (in particular if `opt.autodoc` is `false`), this feature is disabled.

If `opt.autodoc` is a record, it may contain the following entries.

files

A list of files (given by paths relative to the package directory) to be scanned for **AutoDoc** documentation comments. Usually it is more convenient to use `autodoc.scan_dirs`, see below.

scan_dirs

A list of subdirectories of the package directory (given as relative paths) which **AutoDoc** then scans for `.gi`, `.gd`, `.g`, and `.autodoc` files; all of these files are then scanned for **AutoDoc** documentation comments.

Default value: [`"."`, `"gap"`, `"lib"`, `"examples"`, `"examples/doc"`].

level

This defines the level of the created documentation. The default value is 0. When parts of the manual are declared with a higher value they will not be printed into the manual.

gapdoc

This controls whether and how to invoke **GAPDoc** to create HTML, PDF and text files from your various XML files.

The value should be either `true`, `false` or a record. If it is a record or `true` (the latter is equivalent to specifying an empty record), then this feature is enabled. It is also enabled if `opt.gapdoc` is missing. In all other cases (in particular if `opt.gapdoc` is `false`), this feature is disabled.

If `opt.gapdoc` is a record, it may contain the following entries.

main

The name of the main XML file of the package manual. This exists primarily to support packages with existing manual which use a filename here which differs from the default. In particular, specifying this is unnecessary when using scaffolding.

Default value: `PACKAGE_NAME.xml`.

files

A list of files (given by paths relative to the package directory) to be scanned for **GAPDoc** documentation comments. Usually it is more convenient to use `gapdoc.scan_dirs`, see below.

scan_dirs

A list of subdirectories of the package directory (given as relative paths) which **AutoDoc** then scans for `.gi`, `.gd` and `.g` files; all of these files are then scanned for **GAPDoc** documentation comments.

Default value: [`"."`, `"gap"`, `"lib"`, `"examples"`, `"examples/doc"`].

LaTeXOptions

Must be a record with entries which can be understood by `SetGapDocLaTeXOptions` (**GAPDoc: SetGapDocLaTeXOptions**). Each entry can be a string, which will be given to **GAPDoc** directly, or a list containing of two entries: The first one must be the string `"file"`, the second one a filename. This file will be read and then its content is passed to **GAPDoc** as option with the name of the entry.

gap_root_relative_path

Either a boolean, or a string containing a relative path. By default (if this option is not set, or is set to `false`), references in the generated documentation referring to external

documentation (such as the **GAP** manual) are encoded using absolute paths. This is fine as long as the documentation stays on only a single computer, but is problematic when preparing documentation that should be used on multiple computers, e.g., when creating a distribution archive of a **GAP** package.

Thus, if a relative path is provided via this option (or if it is set to `true`, in which case the relative path `../..` is used), then **AutoDoc** and **GAPDoc** attempt to replace all absolute paths in references to **GAP** manuals by paths based on this relative path. On a technical level, **AutoDoc** passes the relative path to the *gaproot* parameter of **MakeGAPDocDoc** (**GAPDoc: MakeGAPDocDoc**)

extract_examples

Either `true` or a record. If set to `true`, then all manual examples are extracted and placed into files `tst/PACKAGENAME01.tst`, `tst/PACKAGENAME02.tst`, ... and so on, with one file for each chapter. For chapters with no examples, no file is created.

As a record, the entry can have the following entries itself, to specify some options.

units

This controls how examples are grouped into files. Recognized values are "Chapter" (default), "Section", "Subsection" or "Single". Depending on the value, one file is created for each chapter, each section, each subsection or each example. For all other values only a single file is created.

On a technical level, **AutoDoc** passes the value to the *units* parameter of **ExtractExamples** (**GAPDoc: ExtractExamples**).

skip_empty_in_numbering

If set to `true` (the default), the generated files use filenames with strictly sequential numbering; that means that if chapter 1 (or whatever units are being used) contains no examples but chapter 2 does, then the examples for chapter 2 are put into the file `tst/PACKAGENAME01.tst`. If this option is set to `false`, then the chapter numbers are used to generate the filenames; so the examples for chapter 2 would be put into the file `tst/PACKAGENAME02.tst`.

4.1.2 InfoAutoDoc

▷ InfoAutoDoc

(info class)

Info class for the **AutoDoc** package. Set this to 0 to suppress info messages, 1 to allow most messages, and 2 to allow all messages including those that contain file paths.

This can be set by calling, for example, `SetInfoLevel(InfoPackageManager, 0)`. Default value 1.

4.2 Examples

Some basic examples for using **AutoDoc** were already shown in Chapter 1.

References

- [LN12] F. Lübeck and M. Neunhöffer. *GAPDoc (Version 1.5.1)*. RWTH Aachen, 2012. GAP package, <http://www.math.rwth-aachen.de/~Frank.Luebeck/GAPDoc/index.html>. 2

Index

`@Arguments` *args*, 13
`@BeginChunk` *name*, 18
`@BeginCode` *name*, 19
`@BeginExample`, 17
`@BeginExampleSession`, 17
`@BeginGroup`, 16
`@BeginLatexOnly`, 19
`@BeginLog`, 18
`@BeginLogSession`, 18
`@BeginNotLatex`, 19
`@Chapter`, 15
`@ChapterInfo`, 14
`@ChapterLabel`, 15
`@ChapterTitle`, 15
`@Description` *descr*, 13
`@DoNotReadRestOfFile`, 18
`@EndChunk`, 18
`@InsertChunk` *name*, 18
`@EndCode`, 19
`@EndExample`, 17
`@EndExampleSession`, 17
`@EndGroup`, 16
`@EndLatexOnly`, 19
`@EndLog`, 18
`@EndLogSession`, 18
`@EndNotLatex`, 19
`@Group` *grpname*, 14
`@GroupTitle`, 16
`@InsertCode` *name*, 19
`@Label` *label*, 14
`@LatexOnly` *text*, 19
`@Level`, 16
`@NotLatex` *text*, 19
`@ResetLevel`, 17
`@Returns` *ret_val*, 13
`@Section`, 15
`@SectionLabel`, 15
`@SectionTitle`, 15
`@Subsection`, 16
`@SubsectionLabel`, 16
`@SubsectionTitle`, 16

Abstract field in `PackageInfo.g`, 9
Acknowledgements field in `PackageInfo.g`, 9
`AProperty`
 for `IsObject`, 14
 `testlabel`, 14
`AutoDoc`, 25
`AutoDocWorksheet`, 24

Bibliography field in `makedoc.g`, 9

Copyright field in `PackageInfo.g`, 9

Entities record in `makedoc.g`, 9

`FirstOperation`
 for `IsInt`, 21

`InfoAutoDoc`, 29

`LaTeXOptions` record in `makedoc.g`, 9

`makedoc.g`, 5

Scaffold record in `makedoc.g`, 8
`SecondOperation`
 for `IsInt`, `IsGroup`, 21

`ThirdOperation`
 for `IsGroup`, `IsInt`, 21